Exercices du bac et traduction pour Xcas

Renée De Graeve
avec la participation de B. Parisse et G. Connan

14 mars 2007

Table des matières

I - Sujet 1

$$u(n+1) = u(n) + a*n + b$$

Trouver $u(n)$ en fonction de n.

a. Avec Xcas

On exécute le fichier bac1_1.xws qui utilise le tableur (formel) et des valeurs particulières, le graphe et un programme que voici :

```
u(n,u0,a,b):={
 local val;
 if (n==0) return u0;
 val:=u0;
 for (k:=1;k<=n;k++) {
  val:=normal(val+(k-1)*a+b);
 }
 return val;
}
```

En faisant ce programme itératif on remarque qu'à la k-ième étape on ajoute à la valeur précedente `b` et `(k-1)*a` car $u(n) = u(n-1) + a*(n-1) + b$.

On ajoute `b` à chaque étape donc, au bout de `n` étapes on aura ajouté `n*b`, et on ajoute `(k-1)*a` à la k-ième étape donc, au bout de `n` étapes on aura ajouté

`1*a+2*a+...+(n-1)*a=(1+2+...+(n-1))*a`

On tape pour avoir la valeur factoriser de $(1+2+...+(n-1))$:

`factor(sum(k,k,1,n-1))`

et on obtient :

`n*(n-1)/2`

On écrit donc la fonction `u` :

`u(n,u0,a,b) :=u0+n*(n-1)*a/2+n*b`

Voir aussi bac1_2.xws qui recherche A, B, C pour que :

$u(n) = A*n^2 + B*n + C$ en résolvant le système linéaire d'inconnues A, B, C :

$u(0) = C$,

$u(1) = A + B + C$,

$u(2) = 4*A + 2*B + C$

b. La démonstration

On a :

$$\begin{array}{rcl}
u(1) & = & u(0) + 0 + b \\
u(2) & = & u(1) + a + b \\
u(3) & = & u(2) + 2*a + b \\
\dots & & \\
u(n) & = & u(n-1) + (n-1)*a + b
\end{array}$$

en ajoutant membre à membre on obtient :

$$u(n) = u(0) + (1+2+...(n-1))*a + n*b$$

donc :

$$u(n) = u(0) + n*(n-1)*a/2 + n*b$$

II - Sujet 2

Dans le plan 4 points O, A, B, C et un cercle E de centre O. À tout point M sur E on associe N tel que :

$$\overrightarrow{MN} = a * \overrightarrow{MA} + b * \overrightarrow{MB} + c * \overrightarrow{MC}$$

où a, b, c sont des réels donnés. Déterminer le lieu de N lorsque M décrit E.

a. Avec Xcas

On exécute la session bac2.xws ou on tape :

```
[a,b,c]:=[1,-3,2];
A:=point(2);
B:=point(1+2*i);
C:=point(-2+i);
O:=point(0,0);
E:=cercle(O,1);
t:=element(0..2*pi);
M:=element(E,t);
N:=point(affixe(M)+a*(A-M)+b*(B-M)+c*(C-M));
affichage(lieu(N,M),rouge+line_width_2);
```

En déplacant le curseur t on déplace M sur le cercleE et on voit N décrire le lieu.
La commande lieu de Xcas nous donne l'équation du lieu, par exemple :

- pour [a,b,c] :=[1,-3,2] l'équation du lieu est :
  4*x^2+40*x+4*y^2+32*y+160=0 qui est le cercle translaté de E dans la translation de vecteur affixe(a*A+b*B+c*C).
  On tape si a+b+c=0 :
  P :=a*A+b*B+c*C;segment(0,P);
  Mt :=affichage(translation(affixe(P),M),quadrant2)
  Mt et N sont alors confondus.
- pour [a,b,c] :=[2,-3,2] N=point(-3-4*i) est confondu avec le barycentre G des points A, B, C pondérés par a, b, c.
  On tape si a+b+c=1 :
  G :=affichage(barycentre([A,a],[B,b],[C,c]),quadrant3) G et N sont alors confondus.
- pour [a,b,c] :=[1,-3,1] l'équation du lieu est :
  16*x^2+96*x+16*y^2+160*y+480=0 qui est l'homothétique de E dans l'homothétie de entre G et de rapport 1-(a+b+c).
  On tape si a+b+c !=0 et a+b+c !=1 :
  Mh :=affichage(homothetie(G,1-(a+b+c),M),quadrant4); Mh et N sont alors confondus.

b. La démonstration

Soit M un point de E et soient P et N définis par :

$$\overrightarrow{OP} = a * \overrightarrow{OA} + b * \overrightarrow{OB} + c * \overrightarrow{OC} \text{ et } \overrightarrow{MN} = a * \overrightarrow{MA} + b * \overrightarrow{MB} + c * \overrightarrow{MC}$$

alors, $\overrightarrow{MN} = (a+b+c) * \overrightarrow{MO} + a * \overrightarrow{0A} + b * \overrightarrow{OB} + c * \overrightarrow{OC}$ donc :

$$\overrightarrow{MN} = (a+b+c) * \overrightarrow{MO} + \overrightarrow{OP}$$

- Supposons $a+b+c=0$, alors $\overrightarrow{MN} = \overrightarrow{OP}$.
  Donc N se déduit de M dans la translation de vecteur $\overrightarrow{OP}$.

- Supposons $a+b+c=1$, soit G le barycentre des points A,B,C pondérés par a,b,c. On a donc $\overrightarrow{MG}(a+b+c)=\overrightarrow{MN}=\overrightarrow{MG}+\overrightarrow{GN}$, soit $\overrightarrow{GN}=\overrightarrow{GM}(1-(a+b+c))=1$ donc N se trouve en G si $a+b+c=1$
- Supposons $a+b+c\neq 0$ et $a+b+c\neq 1$ soit G le barycentre de A,B,C pondéré par a,b,c. On a donc $\overrightarrow{MG}(a+b+c)=\overrightarrow{MN}=\overrightarrow{MG}+\overrightarrow{GN}$, soit $\overrightarrow{GN}=\overrightarrow{GM}(1-(a+b+c))=1$ donc N se déduit de M dans l'homotétie de centre G et de rapport $1-(a+b+c)$.

Application

- a=1,b=-3,c=2
- a=2,b=-3,c=2
- a=1,b=-3,c=1

III - Sujet 3

Soient un rectangle $ABCD$ et un point M à l'intérieur de ce rectangle. Soit H la projection de M sur CD.
Comment choisir M pour que la distance $MA+MB+MH$ soit minimale ?

a. Avec Xcas

Voir `bac3.xws`, la figure est en niveau 1, pour la solution formelle, exécuter les commandes à partir du niveau 2.

b. Solution géométrique



Lorsque M se déplace sur une parallèle à CD la distance $MA+MB$ est minimum lorsque M se trouve en K sur la médiatrice de AB. En effet si $M1$ est le symétrique de M par rapport à K on a $KM = KM1$ et $AM = BM1$ et donc :
$AM+BM = BM+BM1$.
Soit $B1$ le symétrique de B par rapport à K, $MBM1B1$ est un parallélogramme donc : $BM+BM1 = BM1+M1B1 \geqslant BB1 = 2BK$.
Donc $AM+BM \geqslant 2BK = AK+BK$.
On va ensuite montrer que le minimum de $MA+MB+MH$ est lorsque M se trouve au point K définit par : K sur la médiatrice de AB et l'angle K du triangle ABK égal $2\pi/3$.
Si M est sur KH, il suffit de projeter M sur AK en N. N se trouve en dehors de AK car l'angle $\widehat{AKM}$ est obtus et $AN = AK+KN$.
Le triangle NMK est rectangle en N et l'angle K vaut $\pi/3$ donc $KM = 2KN$ et $AN < AM$ c'est à dire : $AN = AK+KN = AK+KM/2 < AM$ soit
$2AK+KM = AK+BK+KM < 2AM = AM+BM$ donc
$AK+BK+KM+MH = AK+BK+KH < AM+BM+MH$.



Si M est sur KI (I milieu de AB), on projette toujours M sur AK en N, mais N se trouve entre A et K car l'angle $\widehat{AKM}$ est aigu et $AK = AN + NK$.
Le triangle NMK est rectangle en N et l'angle K vaut $\pi/3$ donc :
$MK = 2NK$ et $AN < AM$ c'est à dire :
$AN = AK - NK = AK - MK/2 < AM$ soit $AK + BK = 2AK < 2AM + MK = AM + BM + MK$ donc
$AK + BK + KH < AM + BM + MK + KH = AM + BM + MH$.



Lorsque M est en K, AKI est la moitié d'un triangle équilatéral de hauteur $AI = 1/2$ et de côté $AK = 1/\sqrt{3} = \sqrt{3}/3$ ($KI = \sqrt{3}/6$). Donc le minimum de de la longueur $AM + BM + MH$ est : $a + 2 * \sqrt{3}/3 - \sqrt{3}/6 = a + \sqrt{3}/2$.

IV - Sujet 4

Étude, selon les valeurs de m réel donné, du nombre de solutions, pour x strictement positif, de l'équation :

$$(E) \quad \ln(x) = m * x^2$$

a. Avec `Xcas`

Dans `bac4_1.xws` on trace les graphes de $y = \ln(x)$ et de $y = m * x^2$ pour différentes valeurs de m puis, on regarde séparément les cas $m > 0$ et $m < 0$
ou dans `bac4_2.xws` on trace le graphe de la fonction $\ln(x) - kx^2$ pour différentes valeurs de k en faisant une animation.

b. La démonstration

Soit pour x strictement positif $f(x) = \ln(x) - m * x^2$.
On cherche le nombre de solutions de $f(x) = 0$.
Pour cela on va étudier les variations de f.
On a $f'(x) = 1/x - 2 * m * x = (1 - 2 * m * x^2)/x$.
Quand x tend vers 0^+, $f(x)$ tend vers $-\infty$ quelquesoit m.
Quand x tend vers $+\infty$, $f(x)$ tend vers $+\infty$ si $m \leq 0$ et vers $-\infty$ si $m > 0$ (pour $m > 0$ on a $f(x) = x^2 * (\ln(x)/x^2 - m)$ et on sait que $\ln(x)/x^2$ tend vers 0 quand x tend vers $+\infty$).

- Si $m \leq 0$, $f'(x) > 0$ pour x strictement positif donc f est croissante.
  On a :
  quand x tend vers 0^+ $f(x)$ tend vers $-\infty$ quand x tend vers $+\infty$ $f(x)$ tend vers $+\infty$ f est continue et croissante, donc d'aprés le théorème des valeurs intermédiaires, f s'annule une seule fois.
- Si $m > 0$, $f'(x)$ s'annule pour $x = 1/\sqrt{2 * m}$, $f'(x)$ est positive sur $]0; 1/\sqrt{2 * m}[$ et $f'(x)$ est négative sur $]1/\sqrt{2 * m}; +\infty[$ donc f est croissante puis décroissante et son maximum vaut : $f(1/\sqrt{2 * m}) = -1/2(\ln(2 * m) + 1)$.
  Si $\ln(2 * m) + 1 < 0$ c'est à dire si $m < 1/(2 * e)$, f ne s'annule pas et il n'y a pas de solutions,
  Si $\ln(2 * m) + 1 = 0$ c'est à dire si $m = 1/(2 * e)$, f s'annule une seule fois en $x = 1/\sqrt{2 * m} = \sqrt{e}$,
  Si $\ln(2 * m) + 1 > 0$ c'est à dire si $m > 1/(2 * e)$, f s'annule deux fois d'aprés le théorème des valeurs intermédiaires : une fois avant $x = 1/\sqrt{2 * m}$ et une fois aprés.

V - Sujet 5

Étude de la suite `u(n+1)=a*u(n)+b` pour `a` et `b` donnés : convergence et valeur de la limite.

a. Avec `Xcas`

Dans `bac5_1.xws`, on utilise le tableur de `Xcas` au niveau 1 et, au niveau 2, on tape le programme :

```
u(n,u0,a,b):={
 local val;
 if (n==0) return u0;
 val:=u0;
 for (k:=1;k<=n;k++) {
   val:=normal(val*a+b);
 }
 return val;
}
```

En faisant ce programme itératif on remarque qu'à chaque étape : on multiplie la valeur précédente par `a` et on lui ajoute `b` donc, au bout de `n` étapes on aura multiplié la valeur initiale par `a^n` et on aura ajouté la somme :
`b+a*b+.....a^(n-1)*b`
On tape pour avoir la valeur factoriser de $1+a+...+a^{n-1}$:
`factor(sum(a^k,k,1,n-1))`
et on obtient :
`(a^n-1)/(a-1)`
On écrit donc la fonction `u` et on tape :
`u(n,u0,a,b) :=normal(u0*a^n+b*(a^n-1)/(a-1))`
Voir aussi `bac5_2.xws` dans cette session on écrit le programme itératif précédent, puis à l'aide d'exemples et de graphes, on cherche si la suite a une limite pour différentes valeurs de a et b.

b. La démonstration

Si `a=1`, la suite `u` est une suite arithmétique de raison `b`.
Si `a !=1`, soit `l` le rèel vérifiant `l=a*l+b`, c'est à dire `l=b/(1-a)`.
En retranchant membre à membre `l=a*l+b` à `u(n+1)=a*u(n)+b` et en posant `v(n)=u(n)-l`, on obtient : `v(n+1)=a*v(n)`, la suite `v` est donc une suite géométrique de raison `a` donc :
`v(n)=v(0)*a^n=(u(0)-l)*a^n` c'est à dire
`u(n)=v(n)+l=b/(1-a)+(u(0)-b/(1-a))*a^n`
On a donc :
$u(n)=u(0)*a^n+b*(1-a^n)/(1-a)=a^n*(u(0)-\frac{b}{1-a})+\frac{b}{1-a}$
Convergence de u :
Si $abs(a)<1$, alors a^n tend vers 0 quand n tend vers l'infini donc $u(n)$ tend vers $b/(1-a)$.
Si $abs(a)>1$ alors a^n tend vers ∞ donc quand n tend vers l'infini donc $|u(n)|$ tend vers l'infini.
Si $a=1$, $u(n)=u(0)$, donc u(n) est constante.
Si $a=-1$, $u(2n)=u(0)$ et $u(2n+1)=b-u(0)$: si $u(0)=b/2$ la suite $u(n)$ est constante et si $u(0)!=b/(1-a)$ la suite u est divergente.

VI - Sujet 7

Tangente à la courbe $y = \exp(x)$ passant par l'origine : construction et équation.

a. Avec Xcas

Dans bac7.xws on envisage deux points de vue :

– On trace la droite $y = mx$ passant par l'origine. En faisant varier sa pente m on essaye de rendre cette droite, tangente au graphe de $y = \exp(x)$.
  On tape :

```
plot(exp(x));
m:=element(0..5);
droite(y=m*x);
```

  On fait bouger le curseur m et on voit que la droite $y = m * x$ est tangente pour m proche de 2.7 avec le point de tangence de coordonnées $(1;2.7)$.

– On trace la tangente au graphe de $y = \exp(x)$ et passant par le point d'affixe $a + i\exp(a)$. En faisant varier l'abscisse a on essaye de faire passer cette droite par l'origine.
  On tape :

```
plot(exp(x));
assume(a:=-1,-5,5);
T:=tangent(plotfunc(exp(x)),a);
normal(equation(T))
```

  on fait bouger a pour que T passe par O, on voit que a se trouve proche de 1.
  On obtient comme équation de la tangente :

$$y = (-\exp(a) * a + \exp(a) * x + \exp(a))$$

  Cette droite passe par l'origine si et seulement si $-exp(a) * a + exp(a) = 0$.
  Il ne reste donc plus qu'à résoudre $-\exp(a) * a + \exp(a) = 0$.
  On tape `solve(-exp(a)*a+exp(a),a)` et on obtient `[1]`.
  On trouve ainsi que la tangente passe par l'origine lorsque $a = 1$.

b. La démonstration

L'équation de la tangente au graphe de $y = f(x)$ au point $(a; f(a)$ est :

$$y = f'(a) * (x - a) + f(a)$$

Ici on a $f(x) = \exp(x)$, donc $f'x) = \exp(x)$.
La tangente au point d'abscisse a a donc pour équation :

$$y = \exp(a)(x - a + 1)$$

cette droite passe par l'origine si et seulement si $a = 1$ et cette tangente a pour équation $y = \exp(1) * x$.

VII - Sujet 8

Retrouver la troisième loi de Kepler.
On rappelle :
Première loi de Kepler : les planètes décrivent des ellipses dont le soleil occupe un des foyers.
Deuxième loi de Kepler : le rayon vecteur qui joint le soleil à la planète balaie des aires égales en des temps égaux.
Troisième loi de Kepler : les carrés des temps mis par les planètes à parcourir leur orbite (durées de révolution sidérale) sont proportionnels aux cubes des grands axes de ces orbites.

On prend comme unité de longueur le demi-grand axe de la terre et les jours pour la durée.
On donne :

Planète	grand axe	révolution sidérale
Mercure	0.78	88
Vénus	1.44	225
Terre	2	365.25
Mars	3.04	687.25
Jupiter	10.4	4332.75
Saturne	19.1	10759.25
Uranus	38.44	30688.0
Neptune	60.22	60181.0

a. Avec `Xcas`

On ouvre un tableur (voir `bac8.xws`) et on lui donne comme nom `P` puis on le remplit avec ces données :
sur la ligne 0 on met les titres,
dans la colonne A on met le nom des planètes,
on met le grand axe `ga` dans la colonne `B` et,
la durée de la révolution sidérale `dr` dans la colonne `C`.
On trace le nuage de points obtenus, pour cela on tape en dehors du tableur :
`scatterplot(P[1..8,1..2])`
On observe que cette courbe ressemble à une fonction puissance, on met alors dans la colonne `D`, le logarithme de la colonne `B` et dans la colonne `E`, le logarithme de la colonne `C`. On trace le nuage des points définis par les colonnes `D` et `E` et on les relie entre eux par des segments, pour cela on tape en dehors du tableur :
`polygonplot(P[1..8,3..4])`
et le dessin est presque une droite.
On définit les deux points extrémités de ce graphe en tapant :
`A :=point(op(P[1,3..4]));B :=point(op(P[8,3..4]));`

Puis on dessine sur un même graphique la droite (en rouge) passant par ces deux points extrémités et le nuage de points obtenus reliés par des segments :

```
D :=droite(A,B,affichage=rouge);
polygonplot(P[1..8,3..4])
```

On demande ensuite l'équation de cette droite :

```
equation(D)
```

on obtient :

```
y=(1.50185813695*x+4.85049052866)
```

Donc on a approximativement : `ln(dr)=1.5*ln(ga)+4.85`

On tape :

```
dr2 :=exp2pow(exp(3*ln(ga)+4.85 ))
```

on obtient :

```
16317.607198*ga^3
```

Ce qui prouve que l'on a `dr^2=16317.607198*ga^3`.

Dans `F0`, on met `=C0^2/B0^3` et on remplit la colonne `F` avec cette formule : la colonne `F` est presque constante, cette constante est le coefficient de proportionnalité de la troisième loi de Kepler.

Pour avoir une approximation de cette constant on peut demander la moyenne de la colonne `F` en remplisant `G1` par `=mean(F1 :F8)` ou en tapant dans une ligne de commande :

`mean(P[1..8,5])` car `F` est la 5-ième colonne.

On obtient `16653.3941258`

On peut aussi taper :

```
88^2/0.78^3=16675.9453125
225^2/1.44^3=16954.2100694
365.25^2/8=16318.5488629
687.25^2/ 3.04^3=16811.5883079
4332.75^2/10.4^3=16688.8820004
10759.25^2/19.1^3=16613.6055852
30688.0^2/38.44^3=16580.0957393
60181.0^2/60.22^3=16584.2771284
```

VIII - Sujet 10

Un pion se déplace sur l'axe des `x` gradué. Il part de l'abscisse zéro et avant chaque déplacement on lance une pièce de monnaie : si c'est pile le pion recule d'une unité, si c'est face il avance de 2 unités. On effectue 10 tirages soit 10 déplacements.
Quelles sont ses différentes abscisses des points d'arrivée ?
Quelle est la probabilité pour que l'abscisse du pion soit supérieure stricement à 5 unités ?

a. Avec `Xcas`

voir `bac10.xws` On remplit la première ligne du tableur en mettant les différentes abscisses du pion pour cela :
dans `A0` on met `0`, puis dans `B0` on met :
`when(rand(2),A0+2,A0-1)`
`rand(2)` vaut `0` ou `1` : si `rand(2)` vaut `1`, on ajoute `2` à la valeur de la colonne précédente et si `rand(2)` vaut `0` on retranche `1` à la valeur de la colonne précédente. On recopie cette formule sur jusqu'à la colonne `K`.
Puis cette ligne est recopée jusque la ligne `32`.
On obtient ainsi différentes abscisses d'arrivée dans la colonne `K`.
Il faut alors remplir la colonne `L1 :L31` avec les valeurs allant de `-10` à `20` representant à priori les valeurs possibles des abscisses d'arrivée. Puis dans la colonne `M` on compte combien de fois ces abscisses ont été obtenues dans `K0 :K32` en tapant dans `M1` :
`count_eq(L1,K$0 :K$32)` (ne pas oublier $)
formule que l'on recopie dans la colonne `M` jusque `M31`.
Puis on fait le diagramme en batons des colonnes `L1 :L31` et `M1 :M31`
On trace ensuite en rouge et légèrement décalé le graphe théorique correspondant.

b. La démonstration

Au premier déplacement le pion à pour abscisse -1 ou 2, au deuxième déplacement le pion à pour abscisse -2,1 ou 4....Si pour les 10 tirages on a ontenu j piles et k faces ($j+k=10$ ou $k=10-j$ avec $j=0..10$), le pion aura pour abscisse $-k+2*j=3*j-10$ avec $j=0..10$.
Les différentes abscisses des points d'arrivée sont donc :
$-10,-7,-4,-1,2,5,8,11,14,17,20$.
La probabilité pour que l'abscisse du pion soit égale à $3*j-1$ est la probabilité d'avoir j piles pour 10 tirages de pile ou face (probabilité d'avoir pile est $p=1/2$ et probabilité d'avoir face est $q=1-p=1/2$). Cette probabilité est donc :
`binomial(10,j,1/2)=comb(10,j)*(1/2)^10`
Donc la probabilité pour que l'abscisse du pion soit supérieure strictement à 5 unités est :
`sum(binomial(10,j,1/2),j,6,10)`
On obtient : `193/512=0.376953125`
Vérifions :
`sum(binomial(10,j,1/2),j,0,5)`
On obtient : `319/512=0.623046875`
et on a bien `193/512+319/512=1`

IX - Sujet 12

Soit un triangle rectangle direct CBA, d'hypothénuse AB. Lieu de C lorsque A se déplace sur le demi-axe Ox et B se déplace sur le demi-axe Oy.

a. Avec Xcas

Soit un triangle rectangle direct CBA d'hypoténuse AB=1 et d'angle $A = a$. On cherche le lieu de C quand A se déplace sur Ox et B sur Oy.
C est le transformé de B dans la similitude de centre A, de rapport $\cos(a)$ et d'angle $-a$.
On peut choisir différents paramètres pour faire la figure :

– angle t
  Si D est tel que $OADB$ soit un rectangle, et si t est l'angle que fait OD avec Ox alors, l'abscisse de A vaut $\cos(t)$ et l'ordonnée de B vaut $\sin(t)$. Le lieu de C sera obtenu en traçant la représentation paramétrique de l'affixe de C.
– l'abscisse xa de A
  B est alors le point d'affixe $i * \sqrt{1 - xa^2}$.

On choisit de prendre t comme paramètre dans la session `bac12_1.xws` :



On tape :

```
assume(t:=[pi/3,0,pi/2]);
A:=point(cos(t));
B:=point(i*sin(t))
assume(a:=[pi/6,0,pi/2]);
C:=similitude(A,cos(a),-a,B);
plotparam(affixe(C),t);
c1:=re(affixe(C));
c2:=im(affixe(C));
k:=trigsin(simplify(c2/c1));
```

On trouve `k=sin(a)/cos(a)` donc
C se déplace sur la droite $y = x/\tan(a)$ On cherche ensuite comment varie `c1` l'abscisse de `C`, on tape :
`normal(tlin(diff(c1,t)));solve(diff(c1,t),t)`
Après simplification on trouve les bornes du segment lieu de `C` :
`(sin(a)+i*cos(a))*sin(a)` et `sin(a)+i*cos(a)`

On choisit de prendre t comme paramètre dans la session `bac12_2.xws` :

```
assume(a:=[0.282743334,0.0,1.57079632679]);
A:=element(segment(0,1));
xa:=abscisse(A);
B:=point(i*sqrt(1-xa^2));
C:=similitude(A,cos(a),-a,B);;
triangle(A,B,C);
affichage(lieu(C,A),rouge+line_width_2);
```

Il suffit ensuite de faire bouger `A` entre 0 et 1 pour voir `C` décrire le segment de couleur rouge qui est le lieu de `C`.

b. La démonstration

Les points O, A, B, C sont cocycliques : ils sont sur le cercle de diamètre AB car les angles O et C sont droits. Les angles inscrits de sommets O et A qui interceptent le même arc BC sont donc ègaux.
L'angle de sommet A qui intercepte l'arc BC vaut a, donc l'angle de sommet O qui intercepte l'arc BC vaut aussi a : ainsi l'angle que fait OC avec Ox est de $\pi/2 - a$ et C se trouve donc sur la droite d'équation $y = \tan(pi/2 - a) * x$.
Il faut maintenant trouver les positions extrêmes de C qui sont lorsque :
B est en O et lorsque CA est vertical i.e quand $t = \pi/2 - a$.

X - Sujet 13

Soient ABM un triangle, H son orthocentre et une droite d passant par M. Lieu de H lorsque M décrit d.

a. La figure



b. Avec Xcas

Voir bac13.xws. On définit 2 points A et B à coordonnées exactes et une droite d passant par 2 points C et D à coordonnées exactes : ces 4 points ont des coordonnées particulières et seul le point M va pouvoir varier. Pour faire le dessin avec d'autres valeurs, il suffit de changer les coordonnées de A, B, C, D dans les niveau 1,2,3,4 et de modifier correctement les niveaux 12 et 13 (pour cela voir la démonstration ci-après).

```
A:=point(5,0,'affichage'=0);
B:=point(10,0,'affichage'=0);
C:=point(-4,-8,'affichage'=0)::;
D:=point(4,8,'affichage'=0)::;
d:=droite(C,D,'affichage'=0);
assume(t:=[-1.2,-5.0,5.0]);
M:=point(C+t*(D-C));
perpendiculaire(A,B,M),perpendiculaire(M,A,B);
H:=orthocentre(A,B,M);
//H:=inter_unique(perpendiculaire(A,B,M),perpendiculaire(M,A,B));
affichage(plotparam(affixe(H),t),rouge);
triangle(A,B,M);
```

c. La démonstration

On traite ce problème en géométrie analytique :
On suppose $A=(a,0)$, $B=(b,0)$ et d d'équation $y=k*x$ donc $M=(m,k*m)$.
Cherchons les coordonnées de l'orthocentre $H=(h1,h2)$ du triangle ABM : H est sur la perpendiculaire à AB passant par M d'équation $x=m$ et sur la perpendiculaire à $MB=(b-m,-k*m)$ passant par A d'équation $(b-m)(x-a)-k*m*y=0$ donc :
$h1=m$ et $h2=(b-m)(m-a)/(k*m)$
L'équation du lieu est donc $y=(b-x)(x-a)/(k*x)=-x/k+(b+a)/k-ab/(k*x)$
Ce lieu est une hyperbole d'asymptotes $x=0$ et $y=-x/k+(b+a)/k$ passe par les points A et B. On vérifie en tapant :
pour avoir le lieu en rouge :
`affichage(plotfunc((10-m)*(m-5)/(2*m),m),rouge)`
pour avoir l'asymptote $y=-x/k+(b+a)/k$ en vert :
`affichage(droite(y=-x/2+15/2),vert).`

XI - Sujet 15

Dans l'espace soient deux droites *OB* et *AC* non coplanaires. Déterminer le minimum de la distance *MN* lorsque *M* se déplace sur *OB* et *N* se déplace sur *AC*.

a. Avec Xcas

Voir `bac15.xws`

```
O:=point([0,0,0]);
A:=point([3,0,0]);
B:=point([0,4,0]);
C:=point([0,0,2]);
nodisp(d1:=droite(A,C));d1;
nodisp(d2:=droite(O,B));d2;
assume(u:=[0.4,-1.0,1.0]);
assume(v:=[0.3,-1.0,1.0]);
//assume(u:=[1,-5.0,5.0]);assume(v:=[3,-5.0,5.0]);
M:=element(d1,u);
N:=element(d2,v);
P:=parallele(O,d1,d2);
Q:=perpendiculaire(d1,P);//ne marche pas
d3:=inter(P,Q,couleur=rouge);
//d3:=projection(P,d1,couleur=rouge);
nodisp(d:=perpendiculaire_commune(d1,d2));d;
K:=inter_unique(d,d1);
L:=inter_unique(d,d2);
l:=longueur2(K,L);
f:unapply(normal(longueur2(M,N)),u,v);
df:=diff(f(u,v),[u,v])
solve([13*2*u-8,16*2*v-32],[u,v])
f(4/13,1)-l
G:=point(u,v,f(u,v));
```

On obtient : `diff(f(u,v),[u,v])=[13*2*u-8,16*2*v-32]`



b. La démonstration

Considérons la projection orthogonale H, de M sur AC. Dans le plan MAC on a $MH \leq MN$. Soient P le plan parallèle à OB passant par AC, d la projection orthogonale de OB sur P (représentée en noir) et H_0 la projection orthogonale de M sur P.
On a $MH_0 \leq MH \leq MN$, mais malheureusement H_0 n'est pas sur AC.
Peut-on trouver un point M pour que sa projection sur le plan P soit sur AC ? Il suffit pour cela de considérer l'intersection K de d avec AC. L est le projeté de K sur OB et KL est égal à la distance H_0M.

XII - Sujet 21

Méthode d'Euler pour $y' = a * y$ pour a donné.

a. Avec Xcas

Voir bac21.xws
La méthode d'Euler peut être visualisée avec la commande :
`interactive_plotode`.
On tape :
`interactive_plotode(2*y,[t,y])`
La fenêtre `DispG` s'ouvre et en cliquant dans le graphe, on peut voir la solution de l'équation différentielle passant par ce point.
On peut ici, comparer la mèthode d'Euler qui donne `ya` solution approchée de $y' = a * y,\ y(t0) = y0$ pour a donné avec sa solution exacte qui est :
`y(t)=k*exp(a*t)` avec `k=y0*exp(-a*t0)`.
Soit `y(t)=y0*exp(a*(t-t0))`.
La solution approchée `ya` de $y' = a * y, y(t0) = y0$ sur $[t0, t1]$ depend du nombre `n` de points intermédiaires que l'on choisit pour aller de `t0` à `t1`.
`ya(t1,f,t0,y0,n)` renvoie les points d'affixe la valeur approchée de :
`y(t0),y(t0+p)...y(t0+n*p)=y(t1)` pour `p=(t1-t0)/n` et `y` solution de $y' = f(t, y),\ y(t0) = y0$ avec un découpage de $[t0; t1]$ en `n` intervalles égaux de longueur `p`.
On tape :

```
a:=2;
f(t,y):=a*y;
ya(t1,f,t0,y0,n):={
local k,L,p,t,y;
L:=point(t0+i*y0);
t:=t0;
y:=y0;
p:=(t1-t0)/n;
for (k:=1;k<=n;k++) {
y:=y+p*f(t,y);
t:=t+p;
L:=L,point(t+i*y);
}
return L;
};
```

Puis, on trace sur un même graphique, la solution exacte en rouge et les points obtenus avec la méthode d'Euler en vert on tape :

```
affichage(plotfunc(exp(2*(t)),t=0..2),rouge);
affichage(ya(2,f,0,1,50),vert+point_width_2);
plotfield(2*y,[t,y])
```

en changeant le nombre n de points 50 en 10 ou en 100 on verra comment évoluent les points verts.
On trace aussi le champs des tangentes.

XIII - Sujet 25

On considère la suite u définie pour tout entier n, $n \geqslant 1$ par :

$$u_n = \frac{1}{n}\sum_{k=1}^{n} k(k-1)$$

a. Avec `Xcas`

Voir `bac25.xws` et les commentaires de la session.

XIV - Sujet 26

Soient trois points non alignés A, B, C et un réel k de l'intervalle $[-1;1]$. Déterminer le lieu du barycentre G_k du système de points pondérés :

$$\{(A, k^2+1); (B, k); (C, -k)\}$$

lorsque k décrit l'intervalle $[-1;1]$.

a. Avec Xcas

Voir bac26.xws
On a :

$$\overrightarrow{OG} = (k^2+1) * \overrightarrow{OA} + k * \overrightarrow{OB} - k * \overrightarrow{OC})/(k^2+1)) = f(k) + i * g(k)$$

On crée une figure 2d exacte (attention il est important de créer la figure en mode exact). On clique pour définir 3 points A, B, C, puis on dessine la courbe définie par équation paramétrique en tapant (une ligne par niveau) :

```
assume(k=[-0.5,-1,1]);
g:=normal((k^2+1)*affixe(A)+k*affixe(B)-k*affixe(C))/(k^2+1));
G:=point(g);
plotparam(g,k);
G1:=point(subst(g,k=1));
G_1:=point(subst(g,k=-1));
```

Pour montrer que AG est parallèle à BC, on calcule :

```
normal((g-affixe(A))/(affixe(C)-affixe(B)))
```

et on voit qu'il s'agit du réel :

$$-\frac{k}{k^2+1}$$

XV - Sujet 27

Soit un triangle isocèle ABC de sommet A et de perimètre fixé. Trouver le ou les triangle(s) ABC d'aire maximum.

a. Avec Xcas

Voir bac27.xws

```
assume(t:=[2,0.0,5.0]);
B:=point(-t,affichage=quadrant4);
C:=point(t,affichage=quadrant4);
assume(p:=[3,0,10]);
A:=inter_unique(cercle(C,p-t),demi_droite(0,i));
S:=normal(aire(triangle(A,B,C)));
G:=plotfunc(S,t);
M:=point(t+i*S);
s2:=factor(diff(S^2,t));
sol:=solve(s2,t);
SM2:=normal(subst(S^2,t,p/3));
SM:=normal(sqrt(SM2));
triangle(-p/3,p/3,i*p*sqrt(3)/3,couleur=rouge+line_width_3);
```

On obtient l'aire `S=t*sqrt(p^2+(-2*p)*t)` de maximum `SM=sqrt(3)/9*p^2`.

b. La démonstration

L'aire S d'un triangle de coté a,b,c et de demi-périmètre p est :
$S=\sqrt{p(p-a)(p-b)(p-c)}$
ici le triangle ABC est isocèle donc $b=c$ et $p=b+a/2$ ou encore $p-b=p-c=a/2$.
On a donc :
$S^2=pa^2(p-a)/4$.
Le problème revient donc à choisir a pour que S ou S^2 soit maximum, c'est à dire de trouver le maximum de la fonction `f(a) := p*a^2*(p-a)/4`.
On tape : `factor(diff(f(a),a))`
On obtient : `(a*p*(2*p-3*a))/4`
On tape : `solve((a*p*(2*p-3*a))/4,a)`
On obtient :
`[(2*p)/3,0]`
On en déduit que S est maximum quand $a=2p/3$ donc quand $b=c==p-a/2=a=2p/3$ c'est à dire quand le triangle ABC est équilatéral.

XVI - Sujet 29

Pour tout entier n, on définit a et b en posant :

$$a(n) = 4n+1 \text{ et } b(n) = 5n+3$$

Trouver selon n le PGCD$(a(n), b(n))$

a. Avec Xcas

Voir `bac29.xws`
Tout d'abord avec `Xcas` quelque soit n :

```
gcd(4*n+1,5*n+3)=1
```

car `Xcas` calcule le pgcd de 2 polynômes.
On tape pour avoir une idée de la réponse :

```
f(n):=gcd(4*n+1,5*n+3);
f(n)$(n=0..25);
```

On obtient : `(1,1,1,1,1,7,1,1,1,1,1,1,7,1,1,1,1,1,1,7,1,1,1,1,1,1)`
Ce qui veut dire que l'on peut penser que $a = 4n+1$ et $b = 5n+3$ sont premiers entre eux si $n \neq 5+7k$ avec $k \in \mathbb{N}$ et vaut 7 si $n = 5+7k$ avec $k \in \mathbb{N}$.
Ou on utilise le tableur (avec 100 lignes, si on veut, mais est-ce bien utile ?)

b. La démonstration avec l'aide de Xcas

Pour avoir une idée de la démonstration on utilise la fonction `egcd` de `Xcas` qui renvoie les coefficients de l'identité de Bézout pour des polynômes.
On tape :

```
egcd(4*n+1,5*n+3,n);
```

On obtient : `[5,-4,-7]`
Ce qui veut dire que

$$5(4n+1) - 4(5n+3) = -7$$

donc que le PGCD$(4n+1, 5n+3)$ divise 7 c'est à dire vaut 1 ou 7.
Le PGCD$(4n+1, 5n+3)$=7 si et seulement si $4n+1$ est un multiple de 7, donc si il existe $k \in \mathbb{N}$ tel que :

$$4n+1 = 7k \text{ ou encore si } 4n+1 = 0\%7 \text{ donc } 4n = -1\%7 \text{ donc } n\text{=-inv(4\% 7).}$$

On obtient `[-2 % 7]` donc le PGCD de $4n+1$ et $5n+3$ vaut 7 si $n = -2+7k$ avec $k \in \mathbb{N}$ et 1 sinon.

c. La démonstration sans Xcas

On cherche une combinaison linéaire entre $5n+3$ et $4n+1$ qui élimine n.
On a :
$4(5n+3) - 5(4n+1) = 7$
Donc le PGCD$(4n+1, 5n+3)$ divise 7 c'est à dire vaut 1 ou 7.
Le PGCD$(4n+1, 5n+3)$=7 si et seulement si $4n+1$ est un multiple de 7.
Il faut donc trouver n pour avoir $4n+1 = 7k$. Il faut pour cela trouver l'inverse de 4 dans $\mathbb{Z}/7\mathbb{Z}$.
On a : $2*4 = 8 = 1\%7$ donc
$4n = -1 \text{ mod } 7$ est équvalent à $n = -2 \text{ mod } 7$
Donc on trouve que le PGCD$(4n+1, 5n+3)$=7 si $n = -2+7k$ avec $k \in \mathbb{N}$ et que le PGCD$(4n+1, 5n+3)$=1 si $n \neq -2+7k$ avec $k \in \mathbb{N}$.

XVII - Sujet 30

Dans le plan, soient un triangle OAB rectangle en O et une droite d passant par O.
On note A_1 et B_1 les projetés orthogonaux respectifs de A et B sur d.
Étude des cercles de diamètre A_1B_1 lorsque d tourne autour de O.

a. Avec Xcas

Voir `bac30.xws`

```
A:=point(3);
B:=point(i*4);
m:=element(0..pi,pi/3);
d:=droite(y*cos(m)-x*sin(m)=0);
A1:=projection(d,A);
B1:=projection(d,B);
C:=cercle(A1,B1);
C1:=cercle(0,A);
C2:=cercle(0,B);
I:=inter(C1,C2)[0]
```

Le cercle C passe par le point I d'intersection de $C1$ et $C2$ autre que O.

b. La démonstration

$A1$ et $B1$ sont de part et d'autre de O si $\pi/2 < m < \pi$ et du même côté de O si $pi/2 > m > 0$.
En effet, on a les égalités d'angles :
$(BO,BI) = (B_1O,B_1I)$ car cet angle intercepte l'arc OI de $C2$ et
$(AO,AI) + (A_1I,A_1O) = \pi$ car cet angle intercepte l'arc OI de $C1$. $(A_1I,A_1B_1) = (AI,AO)$
$(AI,AO) + (BI,BO) = \pi/2$
$(A_1I,A_1B_1) + (B_1O,B_1I) = \pi/2$
donc A_1I est perpendiculaire à B_1I donc les cercles de diamètre A_1B_1 passent par I.

XVIII - Sujet 33

Dans l'espace, on donne un tétraèdre régulier $OABC$ et I le milieu de AB.
Soit M un point quelconque de AC. Le plan passant par I et orthogonal à IM coupe OB en N. Comment minimiser MN ?



a. Avec Xcas

Voir bac33.xws

```
T:=pyramide([0,0,0],[2,0,0],[0,1,0]);
coordonnees(sommets(T));
A:=point([0,0,0]);
B:=point([2,0,0]);
C:=point([1,sqrt(3),0]);
O:=point([1,sqrt(3)/3,2/3*sqrt(6)]);
I:=milieu(A,B);
assume(t=[0.8,0.0,1.0]);
M:=element(segment(A,C),t);
P:=orthogonal(I,droite(I,M));
N:=inter_unique(droite(O,B),P);
L:=normal(longueur2(M,N));
//M=t*A+(1-t)*C soit M-A=(1-t)*(C-A)
dL:=normal(diff(L,t));
fsolve(numer(dL),t);
subst(L,t,1/2)
```

puis `plotfunc(longueur2(M,N),t);`

b. La démonstration

Pour résoudre ce problème, on veut utiliser différentes méthodes.

1. On utilise les équations des droites et du plan, pour avoir les coordonnés de M et de N.
   On choisit `M :=m+i*m*sqrt(3)` (si `M :=element(segment(A,C),t)` ; on a `m=t`)
   Donc le vecteur IM a pour coordonnées $[m-1, m*\sqrt{3}, 0]$.
   Le plan P a donc pour équation :
   $(m-1)*(x-1)+m*\sqrt{3}*y=0$

et la droite OB a pour équation :
$x = 2-u$,
$y = u/\sqrt{3}$,
$z = 2*u*\sqrt{6}/3$
donc,
N est un point de OB avec $u = 1-m$ i.e. a pour coordonnées :
$[m+1, (1-m)/\sqrt{3}, 2*(1-m)*\sqrt{6}/3]$ soit :
$MN^2 = 1+((1-4*m)*\sqrt{3}/3)^2+8/3*(1-m)^2 = 8*m^2-8*m+4$.
Dans `Xcas` on vérifie en tapant :

```
equation(P);
normal(coordonnees(N));
```

On obtient comme équation de P :
`-t*x-sqrt(3)*t*y+t-(-(sqrt(3)))*y=0`
On obtient comme coordonnées de N :
`[-t+2,sqrt(3)/3*t,6*sqrt(6)/9*t]`
On a comme coordonnées de M :
`[1-t,(1-t)*sqrt(3),0]`
donc,
$MN^2 = (1-t-(-t+2))^2+(\sqrt{3}/3*t-(1-t)*\sqrt{3})^2+(6*\sqrt{6}/9*t)^2 = 8*t^2-8*t+4$.
Le minimum est atteint pour $t = 1/2$ donc pour $m = 1/2$.

2. On calcule MN^2 grâce à des considérations géométriques.
   On choisit `M :=m+i*m*sqrt(3)` ($m = 1-t$)
   Soit K la projection de N sur ABC. K se trouve sur la bissectrice intérieure de l'angle B. L'angle que fait OB avec le plan ABC a pour tangente $\sqrt{2}$ puisque la hauteur h d'un tétraèdre régulier de coté a vaut $h = a*\sqrt{6}/3$. Le sommet O se projette sur ABC en le centre de gravité G du triangle ABC et $GB = a*\sqrt{3}/3$. On a donc $h/GB = a*\sqrt{6}/3/(a*\sqrt{3}/3) = \sqrt{2}$.
   On a donc $NK = BK*\sqrt{2}$.
   Les triangles CMI et BKI sont semblables de rapport $CI/BI = \sqrt{3}$ (ils ont chacun un angle de 30 degrés et leurs angles I sont égaux car ils ont des cotés perpendiculaires), donc : $BK = MC/\sqrt{3} = (2-2*m)/\sqrt{3}$ et $IK = MI/\sqrt{3}$.
   Donc
   $$\begin{aligned} MN^2 &= MK^2+KN^2 = MI^2+IK^2+KN^2 = MI^2*(4/3)+2*BK^2 \\ &= (4*m^2-2*m+1)*4/3+2*(2-2*m)^2/3 = 8*m^2-8*m+4 \end{aligned}$$
   dont le minimum est bien atteint pour $m = 1/2$ donc pour $t = 1/2$.

3. sans faire le calcul de MN^2 en fonction de m.
   On veut minimiser :
   $MN^2 = MI^2*(4/3)+2*BK^2 = (4/3)*MI^2+(2/3)*MC^2$.
   Soit Q la projection de M sur CI.
   On a :
   $MI^2 = MQ^2+QI^2$, $MC^2 = MQ^2+QC^2$ et $MQ = QC/\sqrt{3}$
   donc $MN^2 = 2*MQ^2+(4/3)*QI^2+(2/3)*QC^2 = (4/3)*(QI^2+QC^2)$
   Il faut donc minimiser (QI^2+QC^2). Soit J le milieu de IC (i.e. $\overline{JC} = -\overline{JI}$).
   On a :
   $(QI^2+QC^2) = (\overline{QJ}+\overline{JI})^2+(\overline{QJ}-\overline{JI})^2 = 2(QJ^2+JI^2)$ donc le minimum de (QI^2+QC^2) est obtenu lorsque Q est en J, c'est à dire lorsque M est le milieu de AC.

XIX - Sujet 43

Dans le plan rapporté à un repère orthonormé, on considère :

$$C = \{M(x,y)\ x \geqslant 0,\ y \geqslant 0,\ \sqrt{x} + \sqrt{y} = 1\}$$

C est-il un quart de cercle ?

a. Avec Xcas

Voir bac43.xws

```
plotinequation([x>0,y>0,sqrt(x)+sqrt(y)<1],
               [x=0..1,y=0..1],xstep=0.1,ystep=0.1);
plot((1-sqrt(x))^2,x=0..1);
plotimplicit(sqrt(x)+sqrt(y)-1,
             [x=0..1,y=0..1],xstep=0.05,ystep=0.05);
plot((1-sqrt(x))^2,x=0..1);cercle(1+i,1,affichage=rouge);
```

b. La démonstration

On multiplie par le conjugué, ce qui donne

$$y - x + 1 - 2 * sqrt(y) = 0$$

On remultiplie par le conjugué, on obtient

$$x^2 - 2 * x * y + y^2 - 2 * y + 1 - 2 * x$$

Il reste un terme en xy, donc ce ne peut pas être une équation de cercle.

XX - Sujet 44

Trouver la somme des cubes des n premiers entiers.

a. Avec `Xcas`

On utilise le tableur : voir `bac44.xws`

b. La démonstration

On sait que la somme S_1 des n premiers entiers vaut :
$S_1 = 1+2+...n = n(n+1)/2$
Cherchons la somme S_2 des carrés des n premiers entiers.
On a :
$(1+n)^3 = n^3+3n^2+3n+1$
$n^3 = (n-1)^3+3(n-1)^2+3(n-1)+1$
......
$3^3 = 2^3+3*2^2+3*2+1$
$2^3 = 1^3+3*1^2+3*1+1$
En ajoutant membre à membre, on en déduit que :
$(1+n)^3 = 1^3+3(n^2+(n-1)^2+...2^2+1^2)+3(n+(n-1)+...2+1+n$
$(n^3+3n^2+3n-3n(n+1)/2-n = 3S_2$ Donc la somme des carrés des n premiers entiers vaut : $S_2 = (n^3+3n^2+3n-3n^2/2-5n/2)/3 = (2n^3+3n^2+n)/6 = n*(n+1)*(2*n+1)/6$
Pour calculer la somme des cubes des n premiers entiers on fait la même chose avec $(1+n)^4$: $(1+n)^4 = n^4+4*n^3+6*n^2+4*n+1$ ......
$2^4 = 1^4+4*1^3+6*1^2+4*1+1$
donc $(1+n)^4 = 1^4+4*S_3+6*S_2+4*S_1+n$ donc $4*S_3 = n^4+4*n^3+6*n^2+4*n-(2n^3+3n^2+n)-2n(n+1)-n = n^4+2*n^3+n^2$ donc $S_3 = n^2*(n+1)^2/4$

XXI - Sujet 47

Soit un triangle ABC existe-t-il une droite perpendiculaire à BC qui partage ABC en deux polygônes de même aire ?

a. Avec Xcas

Voir session `bac47.xws`

b. La démonstration

Si $AC = AB$ c'est facile car la hauteur issue de A coupe le triangle isocéle ABC en deux triangles ayant la même aire.



Supposons que $AB > AC$ et soit H le pied de la hauteur issue de A. H peut se trouver entre B et C ou à l'extérieur de BC et du côté de C.
On pose :
$BC = a$, $BH = d$ et $AH = h$
L'aire du triangle ABC est donc : $a * h/2$
On cherche le point M de la droite BC pour que la droite perpendiculaire à BC en M partage ABC en deux polygônes de même aire coupe BC en M.
Soit K le point d'intersection de la perpendiculaire en M à BC avec AB. On a, car les triangles ABH et KMB sont homothétiques :
$BM = x$, $MK = k = h * x/d$ On va considérer trois cas :

- supposons que M est en C
  L'aire du triangle KBC est donc : $a * k/2$ Ce cas se produit si $a * k = a * h * x/d = a * h/2$ c'est à dire si $BC = BM = x = d/2 = BH/2$ Donc si $BC = BH/2$, M est en C

– supposons que M est un point entre B et C i.e. $x < a$.
L'aire du triangle KBM est donc $x*k/2$ et on a $a*k > x*k = a*h/2$ Ce cas peut se produire si $k = h*x/d > h/2$ c'est à dire si $BC > BM = x > BH/2$ et alors on a :
$x*k = x*h*x/d = a*h/2$ c'est à dire $x^2 = a*d/2$. Donc si $BC > BH/2$, M est entre B et C et si on a :
$BM^2 = x^2 = a*d/2$ alors l'aire du triangle BMK est la moitié de l'aire du triangle BCA.
– supposons que M est un point entre C et H. Ce cas peut se produire si $BC < BH/2$.
La perpendiculaire en M à BC coupe AB en K et AC en L.
On a : $MK = k = h*x/d$ et $ML = l = h*(x-a)/(d-a)$ L'aire du quadrilatère $KBCL$ est donc :
$x*k/2-(x-a)*l/2 = (x^2*h/d-(x-a)^2*h/(d-a))/2 =$
$h*(x^2*(d-a)-(x-a)^2*d)/(2*d*(d-a))$ En développant :
$x^2*(d-a)-(x-a)^2*d = -x^2*a+2*x*a*d-a^2*d = -a*(x-d)^2+d*a(d-a)$
on trouve :
l'aire du quadrilatère $KBCL$ est $-h*a*(x-d)^2/(2*d*(d-a))+h*a/2$ M répond à la question si $-h*a*(x-d)^2/(2*d*(d-a))+h*a/2 = a*h/4$
ou encore si :
$-(d-x)^2+d*(d-a)/2 = 0$ ou encore :
$(d-x)^2 = d*(d-a)/2$

c. Construction géométrique de M

– Si $BC \geqslant BH/2$, on détermine M pour que :
$BM^2 = BH*BC/2$.
On trace la perpendiculaire en B à BC et on reporte sur cette perpendiculaire de part et d'autre de B les longueurs $BC/2$ et BH :
$BE = BC/2$ et $BF = BH$
Le cercle de diamètre EF coupe BC en M. Comme le triangle MEF est rectangle en M, on a pour hauteur BM on a :
$BM^2 = BF*BE = BH*BC/2$.
La perpendiculaire en M à BC coupe alors le triangle ABC en deux polygônes de même aire puisque :
le double de l'aire de BMK vaut l'aire de BCA puisque :
$k*x = x^2*h/d = BK*BM/2 = AH*BM^2/BH = AH*BH*BC/(2*BH) = AH*BC/2$
– Si $BC \geqslant BH/2$, on détermine M pour que :
$HM^2 = CH*BH/2$.
On trace la perpendiculaire en H à BC et on reporte sur cette perpendiculaire de part et d'autre de H les longueurs $BH/2$ et CH :
$BE = BC/2$ et $BF = BH$
Le cercle de diamètre EF coupe BC en M. Comme le triangle MEF est rectangle en M, on a pour hauteur HM on a :
$(d-x)^2 = HM^2 = BF*BE = BH*CH/2 = d*(d-a)/2$.
La perpendiculaire en M à BC coupe alors le triangle ABC en deux polygônes de même aire puisque :
le double de l'aire du quadrilatère $KBCL$ est alors égal à : $-h*a*(x-d)^2/(d*(d-a))+h*a = -h*a*d*(d-a)/(2*d*(d-a))+h*a = h*a/2$ Donc M répond bien à la question.